Newsgroup: it.comp.appl.access
Titolo: ACCESS senza ACCESS !!!!!

Salve a tutti!!
Volevo sapere se era possibile avviare una applicazione di ACCESS senza che
sia visibile ACCESS sotto.
Cioč provo a spiegarmi meglio: mi si avvia la mia applicazione con la
maschera che ho definito in avvio, perň vorrei che la "base" sotto di ACCESS
non si vedesse...
Come posso fare???
Grazie a tutti ..
Raffaele

 ha scritto nel messaggio
news:Ly__7.1817$_m5....@news2.tin.it...

Perche' e' brutto e cattivo?



Eliminare completamente Access durante l'esecuzione delle sue
applicazioni, non č possibile. L'unica cosa che puoi fare č
selezionare Avvio... dal menu Strumenti, e disabilitare tutto quello
che non vuoi sia visualizzato durante l'esecuzione dell'applicazione.
Lord Rothgar

 wrote in message
qj2%7.3184$cU4....@twister1.libero.it...

no anzi č utilissimo, perň volevo
che rimanesse un po discretto sotto sotto
a fare il suo bravo lavoro !!!!!!!!!!!!

 wrote in message
gsiq3u4ar3euimf2j...@4ax.com...

Grazie, questa soluzione l'ho sperimentata anch'io, ma volevo
proprio far "sparire" ACCESS, beh pazienza lo tengo cosě
grazie comunque
Raffaele

 Grazie
forse ti e' gia' stato detto, cmq non postare il html.
quello che chiedi e' possibile utilizzando le API, ma _tutte_ le forms
devono
necessariamente essere di tipo PopUp.
inoltre devi prevedere che almeno una form sia sempre aperta, e devi avere
un'
ottima gestione degli errori, perche' altrimenti l'unico modo di uscire dal
programma e' terminarlo!
personalmente ti sconsiglio di procedere su questa strada, ma se proprio
vuoi
provare questo e' il codice:
- per nascondere l'applicazione, da una form di tipo _popup_, esegui:
 fSetAccessWindow(SW_HIDE)
- per riattivarla
 fSetAccessWindow(SW_SHOWNORMAL)
ti ripeto, dalla prima form, puoi aprire solo _form di tipo popup_
in realta' puoi' aprirle tutte, ma non sarebbero visibili
------------------------------------------------------------------------
Public Const SW_HIDE = 0
Public Const SW_SHOWNORMAL = 1
Public Const SW_SHOWMINIMIZED = 2
Public Const SW_SHOWMAXIMIZED = 3
Private Declare Function apiShowWindow Lib "user32" _
    Alias "ShowWindow" (ByVal hwnd As Long, _
          ByVal nCmdShow As Long) As Long
Function fSetAccessWindow(nCmdShow As Long)
'*******************************************
'Name:          fSetAccessWindow (Function)
'Purpose:       Manipulate the main Access window.
'Author:        Dev Ashish
'Date:           04 November 1998
'Called by:     Any
'Calls:           API
'Inputs:         nCmdShow - New state of window
'Output:        True on success, False on failure
'*******************************************
Dim loX  As Long
Dim loForm As Form
    On Error Resume Next
    Set loForm = Screen.ActiveForm
    If Err <> 0 Then 'no Activeform
      If nCmdShow = SW_HIDE Then
        MsgBox "Cannot hide Access unless " _
                    & "a form is on screen"
      Else
        loX = apiShowWindow(hWndAccessApp, nCmdShow)
        Err.Clear
      End If
    Else
        If nCmdShow = SW_SHOWMINIMIZED And loForm.Modal = True Then
            MsgBox "Cannot minimize Access with " _
                    & (loForm.Caption + " ") _
                    & "form on screen"
        ElseIf nCmdShow = SW_HIDE And loForm.PopUp <> True Then
            MsgBox "Cannot hide Access with " _
                    & (loForm.Caption + " ") _
                    & "form on screen"
        Else
            loX = apiShowWindow(hWndAccessApp, nCmdShow)
        End If
    End If
    fSetAccessWindow = (loX <> 0)
End Function
------------------------------------------------------------------------
--
Federico
----------------------------------
Rimuovi paperino. per e-mail
----------------------------------
Tutte le richieste che arrivano
in privato, vengono cestinate.
----------------------------------
Sito Comune di it.comp.appl.access
http://www.sitocomune.com
mailto:
webm...@sitocomune.com
----------------------------------
'------------------- fine ------------------

 sia visibile ACCESS sotto.
Dovresti usare Access ODE ... costa una paccata (a quanto so'), ma sviluppi
applicazioni stand-alone che non necessitano di Access himself.
ciauz,
sorso